

---

Subject: Re: contour plots

Posted by [David Fanning](#) on Mon, 09 Aug 2004 13:15:32 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Hans Clarke writes:

> I also want to know how to transfer the generated plot to a postscript  
> file. Can you help me?

Well, just execute the commands after you have switched  
to the PostScript device driver:

```
thisDevice = !D.Name
Set_Plot, 'PS'
Device, Filename='myplot.ps'

OpenR, lun, 'yourfile.dat', /Get_Lun
rows = File_Lines('yourfile.dat')
header = "
data = FltArr(3,rows-1)
ReadF, lun, header, data
Free_Lun, lun
f = Reform(data[0,*])
period = Reform(data[1,*])
gamma = Reform(data[2,*])
Contour, gamma, period, f, /Irregular, NLevels=10

Device, /Close_File
Set_Plot, thisDevice
```

There are lots of things to know about configuring  
the PostScript device with the DEVICE command, but  
I typically just use my program PSCONFIG to allow  
the user to do it interactively. Saves a lot of  
wear and tear. :-)

```
keywords = PSConfig(Cancel=cancelled)
IF cancelled then RETURN
thisDevice = !D.Name
Set_Plot, 'PS'
Device, _Extra=keywords
```

You can find the programs for PSCONFIG here:

<http://www.dfanning.com/programs/psconfig.zip>

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.dfanning.com/>

---